



# Étude de l'influence de l'enrobage sur la répartition des fissures aux éléments finis avec Cast3M

PIVETEAU Simon

ECAY Lionel

ISABTP - 5<sup>ème</sup> année

8 février 2014

# Sommaire

<b>1</b>	<b>Génération du maillage</b>	<b>3</b>
1.1	Modélisation de la barre HA . . . . .	3
1.2	Modélisation du béton . . . . .	8
1.3	Modélisation de l'arrangement granulaire . . . . .	12
<b>2</b>	<b>Simulation numérique</b>	<b>16</b>
2.1	Combinaison pâte/granulats . . . . .	16
2.2	Conditions aux limites et chargement . . . . .	18
<b>3</b>	<b>Post-traitement des données</b>	<b>20</b>
3.1	Ouverture des fissures . . . . .	20
3.2	Flux des forces à l'interface . . . . .	21

# Introduction

L'objectif de cette étude est de vérifier l'influence de l'enrobage des aciers Haute Adhérence de Béton Armé sur la répartition des fissures. Notre cas d'étude consiste à étudier un morceau de poutre en Béton Armé dont une extrémité est fixe et sur laquelle, à l'autre bout, nous ne tirons que sur l'armature.

À cause de la quantité de ressources de calcul exigée pour faire du calcul plastique, ainsi que pour refaire des calculs avec des enrobages différents, notre étude se limitera à un seul cas de calcul élastique. Si nous avions pour objectif de proposer une loi liant l'espacement des fissures à l'enrobage, il nous faudrait faire une étude statistique sur un grand nombre de tirages de granulats aléatoires, et ce pour chaque rapport enrobage/diamètre d'armature que nous considérerions, ce qui n'était clairement pas concevable dans le cadre de cette initiation au maniement de Cast3M.

# Chapitre 1

## Génération du maillage

Pour générer ce modèle, nous avons considéré un morceau de béton armé de 5,225 cm de haut (axe des  $\vec{y}$ ) par 1,375 cm de large (axe des  $\vec{z}$ ) sur une longueur totale de 15,78 cm (axe des  $\vec{x}$ ). Ce volume comporte deux plans de symétrie, l'un d'eux suivant le plan  $Oxy$  et l'autre suivant le plan  $Oxz$ , de sorte que le calcul représente en réalité une éprouvette 4 fois plus volumineuse que notre échantillon. La première étape consiste à modéliser la barre d'acier.

Sur cette barre de type HA12 (ayant un diamètre de 11 mm) nous trouverons 16 verrous, inclinés à  $48^\circ$  par rapport au plan normal à la ligne moyenne de l'acier. Leur hauteur est de 1 mm et leur longueur de 3 mm en tout, quant à leur espacement, il est de 5,72 mm. Il est à noter qu'ils ont une épaisseur qui varie linéairement de 0 à 1 mm depuis le plan de symétrie  $Oxy$  que nous considérons jusqu'à son point le plus haut.

### 1.1 Modélisation de la barre HA

Nous allons commencer par créer une surface hémicirculaire, que nous déplacerons et tournerons ensuite pour recréer le relief de la barre.

```

----- DÉBUT DU CODE DE CALCUL -----
1 OPTI DIME 3 ELEM TET4 MODE TRID TRAC PSC;
2
3 DENS1=1.E-3;
4 DENSITE DENS1;
5 R=5.5E-3;
6
7 HVERO=1.E-3;
8 ESP_VER=5.72E-3;
9
10 O= 0. 0. 0.;
11 A= 0. R 0.;
```

```

12 B= 0. (-1.*R) 0.;
13
14 S= 0. (R*(2**0.5)/2.) (R*(2**0.5)/2.);
15 T= 0. (R*(2**0.5)/ -2.) (R*(2**0.5)/2.);
16 NVERROU=16;
17
18 L3= CERC A O S C O T C O B;
19 L7=B 3 DROI A;
20 CONT1= L3 ET L7;
21 S1=SURF CONT1 PLANE;
22 CH1=MANU CHPO S1 'SCAL' (R/2.);
23 S1=SURF CONT1 PLANE CH1;

```

---

Déplaçons maintenant cette section vers l'avant et tournons-là, avant de créer le premier verrou.

---

```

24 VECX1=(9.6E-3) 0. 0.;
25 S2=S1 PLUS VECX1;
26
27 ANG0=48;
28 TANG0=TAN ANG0;
29
30 CHX=COORD 2 S2;
31 CHDEP=(EXCO CHX SCAL 'UX') * TANG0;
32 DEPLACER S2 PLUS CHDEP;

```

---

Nous allons maintenant créer le premier verrou. Pour ce faire, il faut savoir que l'excroissance constituant le verrou, vue en coupe, constitue un trapèze axisymétrique, avec une montée, un palier, et une descente. Nous allons déplacer la section que nous venons de tourner d'1 mm en avant, puis créer un champ qui contient les coordonnées en  $\vec{z}$  des éléments de cette section, le normer (en le divisant par sa hauteur maximale) puis le multiplier par la hauteur de verrou. De cette manière, nous créons un champ grâce auquel nous déplacerons ensuite chaque élément d'un certain pourcentage de la hauteur de verrou qui correspond à sa hauteur par rapport au point culminant de notre section. Nous allons ensuite déplacer cette section d'1 mm en avant pour créer le palier puis rattraper la section initiale que nous amènerons de nouveau 1 mm en avant pour matérialiser la descente.

---

```

33 V1=S1 VOLU S2;
34 VECX2=HVERO 0. 0.;
35 S3 = S2 PLUS VECX2;

```

```
36
37 CHZ=COORD 3 S3;
38 ZMAX=MAXI CHZ;
39 CHDEP2=(EXCO CHZ SCAL UZ)*(HVERO/ZMAX);
40 DEPLACER S3 PLUS CHDEP2;
41
42 V2=S2 VOLU S3;
43 S4=S3 PLUS VECX2;
44 V3=S3 VOLU S4;
45 S5=S2 PLUS (3*VECX2);
46 V4=S4 VOLU S5;
47 VECX3=ESP_VER 0. 0.;
48 S6=S5 PLUS VECX3;
49 V5=S5 VOLU S6;
50
51 VTOT1=V1 ET V2 ET V3 ET V4 ET V5;
52 VTOT0=V2 ET V3 ET V4 ET V5;
```

---

Maintenant que nous avons créé le premier verrou ainsi que le premier espacement entre verrous, il ne nous reste plus qu'à automatiser la procédure avec une boucle pour créer l'intégralité des 15 verrous restants. Dans le même temps, nous allons créer un objet séparé contenant la surface extérieure de la barre, qui nous servira plus tard pour créer le volume de béton tout autour.

---

```
53 SBAR00=ENVE V1;
54 SBAR00=DIFF SBAR00 (S1 ET S2);
55 SBAR01=ENVE VTOT0;
56 SBAR02=DIFF SBAR01 (S2 ET S6);
57 SBAR1=SBAR02 ET SBAR00;
58 SPAS=SBAR02;
59 VECX=(3*VECX2) PLUS VECX3;
60
61 REPETER BOU1 (NVERROU - 1);
62   SPAS=SPAS PLUS VECX;
63   SBAR1=SBAR1 ET SPAS ;
64 FIN BOU1;
```

---

Maintenant que nous avons notre barre jusqu'au dernier verrou, il nous faut créer sa face arrière, en déplaçant et en tournant la dernière surface que nous avons créé, tout en ménageant une longueur sans verrou un peu plus importante que de l'autre côté, pour permettre par la suite un glissement de l'acier dans le béton.

---

```

65 S6=S5 PLUS (1.5*VECX1) PLUS ((NVERROU-1) * VECX);
66 CHX=COORD 2 S6;
67 CHDEP=(EXCO CHX SCAL 'UX') * (-1*TANG0);
68 DEPLACER S6 PLUS CHDEP;
69 S7=S5 PLUS VECX3 PLUS ((NVERROU-1) * VECX);
70 V6= S6 VOLU S7;
71 SURFF=ENVE V6;
72
73 SBAR2=SBAR1 ET SURFF;
74 SBAR3=DIFF SBAR2 (S6 ET S7);
75
76 PFAC1=SBAR3 POINT PLAN (0. 0. 0) (0. 1. 0.)
77 (1. 0. 0.) 1.E-5;
78 FAC1=SBAR3 ELEM APPUYE STRICTEMENT PFAC1;
79 SBAR4=DIFF SBAR3 FAC1;

```

---

Nous allons maintenant générer le volume de la barre. Nous allons reprendre le volume du premier verrou créé précédemment et le dupliquer jusqu'à l'autre bout de la barre, puis compléter avec le volume qui redresse la section inclinée.

---

```

80 VPAS=V2 ET V3 ET V4 ET V5;
81 VECX=(3*VECX2) PLUS VECX3;
82
83 REPETER BOU1 (NVERROU - 1);
84   VPAS=VPAS PLUS VECX;
85   VTOT1=VTOT1 ET VPAS;
86 FIN BOU1;
87
88 S7=S6 PLUS (1.*VECX1) PLUS ((NVERROU-1) * VECX);
89 CHX=COORD 2 S7;
90 CHDEP=(EXCO CHX SCAL 'UX') * (-1*TANG0);
91 DEPLACER S7 PLUS CHDEP;
92 VAC=VTOT1 ET V6;
93 ELIM VAC;

```

---

Nous avons maintenant à la fois la surface de notre barre, qui va nous servir de support pour modéliser le béton, et le volume de la barre. Voici le résultat que nous obtenons sur la figure 1.1.



FIGURE 1.1 – Modèle de la barre d'acier haute adhérence

## 1.2 Modélisation du béton

Nous allons maintenant partir du maillage de la barre HA pour réaliser celui du béton qui l'enrobe. Plus précisément, nous allons découper la surface de la barre en trois selon les points créés au tout départ sur notre première section d'acier. Pour pouvoir les distinguer plus aisément, nous allons les colorer en fonction. Puisque les points sont situés à  $\frac{\pi}{4}$  rad et à  $\frac{3\pi}{4}$  rad de l'origine, leurs abscisses sont respectivement de  $\frac{\sqrt{2}}{2}$  et  $-\frac{\sqrt{2}}{2}$ , soit environ 0.717 et  $-0.717$ , c'est pourquoi ces valeurs ont été choisies.

Qui plus est, puisque nous voulons créer un volume d'un matériau différent qui soit au contact de l'autre, il va nous falloir orienter les surfaces de sorte que les deux ne se croisent pas. En effet, le programme a une tolérance de  $10^{-12}$ , ce qui signifie que si l'on crée deux volumes qui ont une arête ou une face en contact (comme c'est le cas ici), lors du calcul le programme placera les points aléatoirement à  $10^{-12}$  près, avec un risque que des points appartenant à l'un des volumes se retrouve placé dans l'autre. C'est à cela que l'opérateur ORIENTER sert, grâce à un vecteur qui lui indique dans quel sens la surface doit être orientée.

---

```

94 ENROBY=6*R;
95 ENROBZ=1.5*R;
96 NZ=ENTIER (ENROBZ/DENS1);
97 NY=ENTIER (ENROBY/DENS1);
98 MOD1=MODE SBAR4 MECANIQUE;
99 CH1=MANU CHML MOD1 SCAL 1. GRAVITE;
100 CHX CHY CHZ=COORD CH1;
101 CHCOS=CHY/(((CHZ**2)+(CHY**2))**0.5) ;
102 VEC0=0. 0. 0.;
103 SBAR41=CHCOS ELEM COMPRIS 0.707 1. COUL BLEU ORIEN (0. 1. 0.) ;
104 SBAR42=CHCOS ELEM COMPRIS -1. -0.707 COUL ROUG ORIEN (0. -1. 0.);
105 SBAR43=DIFF SBAR4 (SBAR41 ET SBAR42) COUL VERT ORIEN (0. 0. 1.);
106 SBAR41B=SBAR41 PLUS VEC0 ORIEN (0. -1. 0.);
107 SBAR42B=SBAR42 PLUS VEC0 ORIEN (0. 1. 0.);
108 SBAR43B=SBAR43 PLUS VEC0 ORIEN (0. 0. -1.);
109 SBAR4B = SBAR41B ET SBAR42B ET SBAR43B;

```

---

Le résultat de ce découpage est visible sur la figure 1.2.

Maintenant que notre maillage est découpé, il va s'agir de projeter chacune des trois surfaces sur les plans qui matérialisent les limites du volume de notre éprouvette.

---

```

110 *
111 * PLAN XZ COTE 1
112 *
113 PXZ1=0. (ENROBY + R) 0.;

```




FIGURE 1.2 – Découpage du maillage en trois zones

```

114 PXZ2=1. (ENROBY + R) 0.;
115 PXZ3=0. (ENROBY + R) 1.;
116 SXZ1=SBAR41B PROJ (0. 1. 0.) 'PLAN' PXZ1 PXZ2 PXZ3;
117 VV1=VOLU NY SBAR41B SXZ1;
118 *
119 * PLAN XZ COTE 2
120 *
121 PXZ21=0. (0.-R-ENROBZ) 0.;
122 PXZ22=1. (0.-R-ENROBZ) 0.;
123 PXZ23=0. (0.-R-ENROBZ) 1.;
124 SXZ2=SBAR42B PROJ (0. -1. 0.) 'PLAN' PXZ21 PXZ22 PXZ23;
125 VV2=VOLU NZ SBAR42B SXZ2;
126 *
127 * PLAN XY
128 *
129 PXY1=0. 0. (ENROBZ + R);
130 PXY2=1. 0. (ENROBZ + R);
131 PXY3=0. 1. (ENROBZ + R);
132 SXY=SBAR43B PROJ (0. 0. 1.) 'PLAN' PXY1 PXY2 PXY3;
133 VV3=VOLU NZ SBAR43B SXY;

```

---

Grâce à ceci, nous obtenons maintenant le volume de la figure 1.3

Il ne nous reste maintenant plus qu'à combler les zones vides entre les trois projections. Pour ce faire, la technique utilisée est de créer en premier lieu un champ par élément sur la surface périphérique de la projection verte, dont nous nous servirons pour en isoler les faces latérales. Cette séparation est faite en isolant les éléments dont




FIGURE 1.3 – Création de 3 volumes par projection

la coordonnée sur l'axe  $\vec{y}$  est la plus faible, pour l'une des deux faces, et ceux dont la coordonnée sur l'axe  $\vec{y}$  est la plus forte, pour la face opposée. Il ne reste par la suite plus qu'à projeter ces surfaces sur les plans délimitant notre éprouvette, comme tout à l'heure, et à éliminer les points en doublon pour recréer l'unité de la matière.

---

```

134 F33=FACE 3 VV3;
135 MOD33=MODE F33 MECANIQUE;
136 CH33=MANU CHML MOD33 'SCAL' 1. GRAVITE;
137 CHY33=COORD 2 CH33;
138
139 YMIN=MINI CHY33;
140 S33M=CHY33 ELEM INFE (YMIN + 1.E-6);
141 SXZ22=S33M PROJ (0. -1. 0.) 'PLAN' PXZ21 PXZ22 PXZ23;
142 VV22=VOLU NZ S33M SXZ22;
143
144 YMAX=MAXI CHY33;
145 S33P=CHY33 ELEM SUPE (YMAX - 1.E-6);
146 SXZ12=S33P PROJ (0. 1. 0.) 'PLAN' PXZ1 PXZ2 PXZ3;
147 VV21=VOLU NY S33P SXZ12;
148
149 VBET=VV1 ET VV21 ET VV3 ET VV22 ET VV2 COUL BLEU;
150 ELIM VBET;

```

---

Cette manipulation nous donne le volume final, tel que montré sur la figure 1.4

Nous allons, avant de passer à la suite, créer les objets contenant les surfaces auxquelles nous imposerons des conditions aux limites pour matérialiser les blocages




FIGURE 1.4 – Création du volume final de béton

de notre système, tant dans le béton que dans l'acier. Pour ce faire, nous utilisons la même astuce que juste au-dessus, en isolant les faces dont les coordonnées sur l'axe  $\vec{x}$  sont de 0 ou de la valeur maximale possible dans notre éprouvette, ce qui se traduit comme suit :

---

```

151 SBET=ENVE VBET;
152
153 MOD1=MODE SBET MECANIQUE;
154 CH1=MANU CHML MOD1 'SCAL' 1. GRAVITE;
155 CHX=COORD 1 CH1;
156
157 XMAX=MAXI CHX;
158 SBX0=CHX ELEM INFE 1.E-6;
159 SBX1=SBX0 PLUS (XMAX 0. 0.);
160
161 ELIM VBET SBX1;
162
163 SAX0=S1;
164 SAX1=SAX0 PLUS (XMAX 0. 0.);
165
166 ELIM VAC SAX1;
167
168 TOUT=VBET ET VAC;
169 ELIM VBET VAC;

```

---

Nous allons sans plus tarder affecter des modèles de comportement et des maté-

riaux aux volumes que nous venons de créer, car ils nous serviront par la suite. Le modèle du béton est un modèle qui tient compte de l'endommagement du matériau car c'est là ce qui nous intéresse, tandis que celui de l'acier reste un modèle élastique classique. Si le matériau affecté au béton à des caractéristiques surprenantes, c'est simplement qu'elles seront mises à jour un peu plus loin et que pour le moment il est avantageux d'avoir un valeur unitaire de module d'Young dont nous allons pouvoir nous servir dans diverses phases de calcul de l'incorporation des granulats.

---

```

170 MOBET= MODE VBET MECANIQUE ELASTIQUE ENDOMMAGEMENT MICROISO;
171 MOAC=MODE VAC MECANIQUE ELASTIQUE;
172 MABET=MATE MOBET YOUN 1. NU 0.18;
173 MAAC=MATE MOAC YOUN 2.1E11 NU 0.25;
174 MOTOT=MOBET ET MOAC ;
175 MATOT=MABET ET MAAC ;

```

---

### 1.3 Modélisation de l'arrangement granulaire

Dans le but d'obtenir un modèle le plus réaliste possible, qui puisse permettre de réaliser des calculs de fissuration, il est très intéressant de procéder à un tirage granulaire aléatoire. C'est à l'aide d'un code en fortran — code extrêmement gourmand en ressources que notre ordinateur était hélas incapable de faire tourner, soit dit en passant — que celui-ci est effectué. Nous avons donc récupéré les données gracieusement fournies par M. LA BORDERIE.

L'étape suivante consiste donc à remplir notre volume de béton avec les granulats du tirage aléatoire. Pour ce faire nous allons commencer par récupérer les données du tirage, et en particulier les coordonnées des centres de granulats ainsi que les rayons qui leur correspondent. Nous allons stocker dans des listes ces valeurs, par l'intermédiaire de boucles qui se répèteront autant de fois qu'il y a de granulats dans les fichiers de données. Chose assez remarquable, dans un volume aussi faible que celui que nous étudions, il rentre pas moins de 12243 granulats.

---

```

176 PROGX=PROG;
177 PROGY=PROG;
178 PROGZ=PROG;
179
180 OPTI ACQU 'XCENTRES.DAT';
181 ACQU NGR*ENTIER;
182 REPETER BOU1 NGR;
183   ACQU X*FLOTTANT ;
184   PROGX=INSE PROGX &BOU1 X;
185 FIN BOU1;

```

```

186
187 OPTI ACQU 'YCENTRES.DAT';
188 ACQU NGR*ENTIER;
189 REPETER BOU1 NGR;
190   ACQU Y*FLOTTANT ;
191   PROGY=INSE PROGY &BOU1 Y;
192 FIN BOU1;
193
194 OPTI ACQU 'ZCENTRES.DAT';
195 ACQU NGR*ENTIER;
196 REPETER BOU1 NGR;
197   ACQU Z*FLOTTANT ;
198   PROGZ=INSE PROGZ &BOU1 Z;
199 FIN BOU1;
200
201 OPTI ACQU 'RAYONS.DAT';
202 ACQU NGR*ENTIER;
203 PROGR=PROG;
204 REPETER BOU1 NGR;
205   ACQU R*FLOTTANT ;
206   PROGR=INSE PROGR &BOU1 R;
207 FIN BOU1;
208
209 NG=DIME PROGR;
210 VP=VBET;

```

---

Maintenant que nos données sont chargées, nous allons pouvoir commencer à les exploiter. L'étape suivante consiste à incorporer les granulats au modèle de béton existant. L'idée ici est de faire un masque qui occultera les zones non comprises dans les granulats et leur assignera la valeur 0, tandis que les zones situées dans les granulats, elles, vaudront 1 (C'est un booléen).

Pour ce faire, nous commencerons par extraire divers champs par élément, contenant respectivement les coordonnées en  $\vec{x}$ ,  $\vec{y}$  et  $\vec{z}$ , ainsi qu'un champ unitaire qui servira de base de calcul.

Nous lancerons ensuite une boucle sur le nombre total de granulats qui va réaliser les étapes suivantes pour chaque granulat :

- Extraire les coordonnées du centre du granulat ainsi que son rayon,
- Faire des champs par élément sur chaque coordonnée qui soient centrés sur l'origine de la sphère représentant le granulat,
- Calculer le carré de la distance à l'origine du champ ainsi reconstitué avec le changement de base que nous venons d'opérer,
- Créer un masque tel que pour toute valeur de ce champ inférieure au carré du rayon du granulat, la condition soit vérifiée et la valeur de l'élément dans le

- masque mise à 1, et que pour tout autre élément la valeur soit mise à zéro,  
 — Effacer toutes les variables intermédiaires pour pouvoir faire le calcul du granulat suivant sans soucis.

Il conviendra de sommer tous les champs par élément ainsi obtenus, ce qui nous donnera le masque obtenu dans la figure 1.5. Par souci d'économie de temps s'il faut refaire des calculs ou manipuler de nouveau les maillages par la suite, nous sauvegarderons ensuite l'ensemble des données collectées dans un fichier de sauvegarde. Pour information, il occupe 28Mo.

---

```

211 CHAMX CHAMY CHAMZ=COORD MABET;
212 CHAM1= EXCO MABET YOUN SCAL;
213 REPETER BOU1 NG;
214   MESSAGE 'GRANULATS #' &BOU1 ' SUR ' NG;
215   X0=EXTR PROGX &BOU1;
216   Y0=EXTR PROGY &BOU1;
217   Z0=EXTR PROGZ &BOU1;
218   R1=EXTR PROGR &BOU1;
219 *
220 * ON DÉFINIT UN CHAMP PAR ÉLÉMENT DÉCRIVANT LE GRANULAT
221 *
222   CHAMX2=CHAMX - (X0*CHAM1);
223   CHAMY2=CHAMY - (Y0*CHAM1);
224   CHAMZ2=CHAMZ - (Z0*CHAM1);
225   CHAMD=(CHAMX2 ** 2) + (CHAMY2 ** 2) + (CHAMZ2 ** 2);
226   OUBL CHAMX2;
227   OUBL CHAMY2;
228   OUBL CHAMZ2;
229 *
230 * ON CHERCHE LES ÉLÉMENTS SITUÉS DANS LA SPHÈRE
231 *
232   CHAMG=CHAMD MASQUE INFERIEUR (R1**2);
233   OUBL CHAMD;
234   SI (&BOU1 EGA 1);
235     CHAMT=CHAMG;
236   SINON;
237     CHAMT=CHAMT + CHAMG;
238   FINSI;
239   OUBL CHAMG;
240   FIN BOU1;
241   OPTI SAUV 'MAILLAGE_MESO.SAUV';
242   SAUV;

```

---




FIGURE 1.5 – Masque présentant les granulats

## Chapitre 2

# Simulation numérique

Pour pouvoir calculer les contraintes, déformations et autres données que nous recherchons, il nous faut maintenant définir des modèles de comportement et des matériaux. Nous allons par conséquent nous baser sur les maillages que nous venons de créer pour affecter les données que nous voulons.

### 2.1 Combinaison de la pâte et des granulats dans un même matériau

Le modèle utilisé est le modèle d'endommagement de M. MAZARS implémenté dans le code de calcul de Cast3M, et il a besoin de divers coefficients pour pouvoir fonctionner. Ceux-ci gouvernent l'allure des courbes de contrainte-déformation dans le cas de la traction comme de la compression, mais seule la traction nous intéresse ici.

Nous allons créer deux jeux de variables que nous affecterons séparément à la matrice cimentaire et aux granulats. De manière générale, les granulats présentent une meilleure résistance et de meilleures caractéristiques mécaniques que la pâte de ciment, c'est pourquoi c'est cette dernière qui sera la plus endommagée et qui présentera de la fissuration. Le jeu de caractéristiques 0 sera attribué aux granulats tandis que le jeu de caractéristiques 1 sera attribué à la pâte de ciment.

---

```
243 YG1=25.E9;
244 FT1=3.E6;
245 GF1=20.;
246 MP1=(10*YG1);
247 COEF2=(YG1/(YG1 + MP1));
248 COEF1=(1.D0-COEF2);
249
250 YG0=35.E9;
251 FT0=6.E6;
```

```

252 GF0=60.;
253 MP0=(10*YG0);
254 COEF2=(YG0/(YG0 + MP0));
255 COEF0=(1.D0-COEF2);
256
257 EPSD1=FT1/YG1;
258 EPSD0=FT0/YG0;

```

---

Maintenant que ces données sont posées, nous allons créer deux matériaux différents sur le même maillage de base.

---

```

259 CHAM1=MANU 'CHML' MOBET BTRA 1. TYPE CARACTERISTIQUES RIGIDITE;
260 CHAM2=INTG CHAM1 MOBET ELEM BTRA;
261 CHH=CHAM2**(1./2.);
262 CHAM3= CHH * (FT0 / GF0/COEF0) ;
263 MATE0=MATE MOBET YOUN YG0 NU 0.2 KTR0 EPSD0
264     BTRA CHAM3 ALF1 60.
265     RHO 2500. MP MP0 FPC 50.E+8 ;
266
267 CHAM1=MANU 'CHML' MOBET BTRA 1. TYPE CARACTERISTIQUES RIGIDITE;
268 CHAM2=INTG CHAM1 MOBET ELEM BTRA;
269 CHH=CHAM2**(1./2.);
270 CHAM3= CHH * (FT1 / GF1/COEF1) ;
271 MATE1=MATE MOBET YOUN YG1 NU 0.2 KTR0 EPSD1
272     BTRA CHAM3 ALF1 60.
273     RHO 2500. MP MP1 FPC 50.E+8 ;

```

---

Maintenant que ces matériaux sont créés, il va nous falloir les assigner au bons éléments. Pour ce faire nous allons extraire un par un les différents composants des matériaux (module d'Young,  $\nu$ ,  $\rho$ , etc...) et les affecter soit au maillage correspondant aux granulats (nommé CHAMT) soit à celui correspondant à la pâte de ciment (CHAM1 - CHAMT). Nous allons ainsi reconstituer des matrices de matériaux « à la main », ce qui nous permettra de recréer un objet MATERIAU unique pour le béton qui regroupe les granulats et la pâte.

---

```

274 CHAM1=EXCO CHAM1 BTRA SCAL;
275 LM2=EXTR MATE0 COMP;
276 NCOMP=DIME LM2;
277 LMO1=MOTS 'SCAL';
278
279 REPETER BOUCOMP NCOMP;

```

```

280 MO2=EXTR LM2 &BOUCOMP;
281 LMO2=MOTS MO2;
282 MA0=EXCO MATE0 MO2;
283 MA1=EXCO MATE1 MO2;
284 MATOT1=(CHAMT * MA0 LMO1 LMO2 LMO2) ;
285 MATOT2=((CHAM1 - CHAMT) * MA1 LMO1 LMO2 LMO2);
286 SI (&BOUCOMP EGA 1);
287 MABET=MATOT1 + MATOT2;
288 SINON;
289 MABET=MABET ET (MATOT1 + MATOT2);
290 FINSI;
291 FIN BOUCOMP;
292
293 MOTOT=MOBET ET MOAC;
294 MATOT=MABET ET MAAC;

```

---

## 2.2 Conditions aux limites et chargement

Maintenant, il nous faut déclarer des conditions aux limites pour pouvoir faire notre calcul. Nous allons bloquer une extrémité de l'acier et du béton tandis qu'à l'autre extrémité nous ne bloquerons que l'acier afin de pouvoir créer notre chargement par la suite.

Nous devons également imposer deux conditions de symétrie sur les plans que nous citons au 1.

```

295 CL1=BLOQ (SAX0 ET SBX0) UX;
296 CL2=BLOQ SAX1 UX;
297 CL3=SYMT DEPL TOUT (0. 0. 0.) (1. 0. 0.) (0. 1. 0.);
298 CL4=SYMT DEPL TOUT PXZ1 PXZ2 PXZ3;
299 CLTOT=CL1 ET CL2 ET CL3 ET CL4;

```

---

Il faut à présent définir un déplacement imposé sur la condition à la limite bloquant l'acier seul CL2 pour simuler la traction sur la barre HA que nous étudions. Il faut ensuite créer un chargement en déplacement imposé pour pouvoir créer la table de la procédure PASAPAS qui fera le calcul.

```

300 F1=DEPI CL2 1.E-04;
301 PROG1=PROG 0. 1.;
302 PROGMEC=PROG 0. 1.;
303 EVOLMEC=EVOL MANU PROG1 PROGMEC;
304 CHARMEC=CHARGEMENT EVOLMEC F1 DIMP;

```

---

Une fois cette étape effectuée, nous pouvons enfin déclarer la table de calcul PASAPAS, puis lancer la procédure et enfin sauvegarder les données obtenues après calcul.

Si le pas de temps choisi est plus grand que celui qui nous avait été communiqué et le nombre de pas de temps plus faible, c'est simplement parce que notre ordinateur n'acceptait pas de calculer avec un pas de temps trop petit, il renvoyait systématiquement une erreur après la première itération. Même avec cette réduction drastique de la quantité de calculs à effectuer, il nous aura fallu pas moins de dix heures de calculs pour parvenir à un résultat.

---

```
305 TAB1=TABLE;  
306 TAB1.MODELE=MOTOT;  
307 TAB1.CARACTERISTIQUES=MATOT;  
308 TAB1.CHARGEMENT=CHARMEC;  
309 TAB1.BLOCAGES_MECANIKES=CLTOT;  
310 TAB1.TEMPS_CALCULES=PROG 0. 1. PAS 1.E-4 NPAS 20;  
311 PASAPAS TAB1;  
312 OPTI SAUV 'TIRANT_MESO.SAUV';  
313 SAUV;
```

---

FIN DU CODE DE CALCUL

---

# Chapitre 3

## Post-traitement des données

### 3.1 Ouverture des fissures

La procédure PASAPAS donne beaucoup de données que l'on peut exploiter. J'ai choisi dans un premier temps d'utiliser la procédure de calcul d'ouverture de fissures OUVFISS développée par MM. LA BORDERIE, MATALLAH et MAUREL. La première chose à faire pour pouvoir le faire est de restaurer les résultats sauvegardés. On peut ensuite tracer les contraintes, les ouvertures de fissures et bien d'autres choses.

On observe sur la figure 3.1 que l'espacement des fissures est plus important sur la face la plus éloignée de l'acier HA. Ce phénomène peut être accentué par le fait que, l'enrobage étant assez faible, les plus gros granulats n'ont pas la possibilité de passer sous la barre, et par conséquent les fissures peuvent déboucher à intervalles plus faibles sur la peau du béton. De manière générale, on observe bien sur la figure 3.1 une localisation de la fissuration à l'intérieur de la pâte quasiment exclusivement.




FIGURE 3.1 – Ouverture de fissures dans le béton

FIGURE 3.2 – Contraintes longitudinales  $\sigma_{xx}$ 

Si l'on regarde les états de contrainte, on s'aperçoit que dans tous les cas les granulats sont plus fortement chargés que la pâte, comme en témoigne la figure 3.2.

### 3.2 Évaluation du flux d'effort transitant par l'interface

Néanmoins, il y a d'autres moyens encore d'exploiter les résultats obtenus. Tout commence là encore par une restauration des données sauvegardées à l'issue du calcul PASAPAS. Nous découperons ensuite notre éprouvette suivant l'axe  $\vec{x}$  en 100 intervalles.

---

DÉBUT DU POST-TRAITEMENT

---

```

314 OPTI REST 'TIRANT_MESO.SAUV';
315 REST;
316
317 X = COORD 1 SAX0;
318 XMIN = MINI X;
319 X = COORD 1 SAX1;
320 XMAX = MAXI X;
321
322 N1 = 100;
323 XPAS = (XMAX-XMIN)/N1
324 VECX = XPAS 0. 0.;
325
326 N2 = DIME (TAB1.TEMPS)

```

---

Nous allons tracer l'évolution des efforts transitant par l'interface acier/béton grâce à une double boucle. La boucle principale répètera l'opération pour chacun des pas de temps calculés par la procédure PASAPAS. Si l'on rentre dans cette boucle, il apparaît clairement que le but est de projeter les contraintes sur la surface de l'interface puis de les intégrer pour en déduire des efforts.

Pour ce faire, nous allons commencer par faire une première projection « à la main » puis ajouter le résultat obtenu ainsi que l'abscisse auquel il l'a été à deux listes séparées. C'est là que la deuxième boucle intervient pour nous permettre d'automatiser la tâche et de continuer ainsi pour autant de surfaces que l'on en veut, pour autant que l'on ait la patience nécessaire.

Finalement, nous pourrions tracer une évolution pour chaque pas de temps, et même les superposer toutes pour voir l'évolution de la quantité d'efforts transitant à l'interface acier/béton à la fois dans l'espace et dans le temps.

---

```

327 REPETER BOUEXT N2;
328 I=&BOUEXT - 1;
329 MOD4 = MODE SAX0 MECANIQUE;
330 SIG1 = TAB1.CONTRAINTE.S.I REDU MOAC CHAN NOEUD MOAC PROI MOD4;
331 FX1 = INTG SIG1 MOD4 SMXX;
332 PROGX = PROG 0.;
333 PROAFX = PROG FX1;
334 REPETER BOU1 N1;
335 SB1 = SAX0 PLUS (&BOU1 * VECX);
336 MOD4 = MODE SB1 MECANIQUE;
337 SIG1 = TAB1.CONTRAINTE.S.I REDU MOAC CHAN NOEUD MOAC PROI MOD4;
338 FX1 = INTG SIG1 MOD4 SMXX;
339 PROGX = PROG 0.;
340 PROAFX = PROG FX1;
341 FIN BOU1;
342 TITRE 'INCREMENT NUMERO' I;
343 EVFX = EVOL MANU PROGX PROAFX;
344 DESS EVFX NCLK;
345 SI (I EGA 0);
346 EVTOT = EVFX;
347 SINON;
348 EVTOT = EVTOT ET EVFX;
349 FINSI;
350 FIN BOUEXT;
351 DESS EVTOT;

```

---

FIN DU POST-TRAITEMENT

---

Au terme de tout ceci, nous obtenons le graphe 3.3.




FIGURE 3.3 – Évolution de la quantité d’efforts transitant à l’interface acier/béton, le long de l’éprouvette et dans le temps

Nous voyons bien sur ce graphe que, puisque nous tirons sur l’acier, le transfert d’efforts est très important sur la première partie de l’éprouvette (le pic est à 10,2 kN). Au fur et à mesure que l’on s’éloigne du point où l’on tire, cependant, la valeur tend vers une asymptote horizontale (7,3 kN). En effet l’effort de traction se répartit dans le matériau et tend à s’homogénéiser.

# Conclusion

Au terme de cette étude, il apparaît clairement que la quantité de ressources informatiques nécessaire à l'établissement d'une loi fondée sur des statistiques de tirages aléatoires permettant de donner une idée de l'espacement des fissures en fonction de l'enrobage est proprement astronomique.

Dans le cadre de notre étude, nous avons pu mettre en évidence certains phénomènes intéressants, comme la localisation de l'endommagement majoritairement dans la pâte, ou encore la concentration des efforts à l'interface plus grande sur les trois premiers centimètres à partir du point où l'on tire sur la barre. Nous avons également pu observer que la fissuration était plus rapprochée là où l'enrobage était le plus faible, et c'est certainement corrélé au fait que là où il y a peu d'enrobage les granulats les plus gros ne peuvent pas se placer sous l'armature.

C'était par ailleurs pour nous un projet très instructif et intéressant, qui nous a permis de voir plus concrètement à quoi ressemble le monde de la recherche.

# Table des figures

1.1	Modèle de la barre d'acier haute adhérence . . . . .	7
1.2	Découpage du maillage en trois zones . . . . .	9
1.3	Création de 3 volumes par projection . . . . .	10
1.4	Création du volume final de béton . . . . .	11
1.5	Masque présentant les granulats . . . . .	15
3.1	Ouverture de fissures dans le béton . . . . .	20
3.2	Contraintes longitudinales $\sigma_{xx}$ . . . . .	21
3.3	Évolution du transit d'efforts à l'interface . . . . .	23